
ABSTRACT
Using Proc Reports to Create HTML pages via % OUT2HTM is very
easy. This paper describes how Suntrust Bank utilizes  this SAS tool
to provide its internal customers from over 3600 Bank Branches with
product information via the WEB.

INTRODUCTION
The use of Proc Reports is widely accepted as an excellent report
generator. Many SAS programmers have wanted to create these
reports out on the WEB. The publication of these reports on the
Intranet has enabled the business analysts to quickly review the
results.
Suntrust Bank developed the weekly open/closed report to closely
monitor its growing deposits. The report provides information on
deposit account activity from a holding company view at its highest
level to a branch view at its lowest level.  This type of information will
help management understand the weekly trends and performance in
deposit product balances. Branch managers can review the state of
deposits for their holding company, state, region, division, state, and
branches.
Version 8  of SAS has an HTML formatting macro called %OUT2HTM
to produce these Reports out to the WEB without having to know a
large amount of HTML(Hypertext Markup Language).  This paper
focuses on how Suntrust Bank created over 3600  branch reports on
the Intranet, as well as, little HTML tidbits that can make the report
more attractive.

HTML FORMATING TOOLS

                                         PICTURE 1

                                         PICTURE 2

HTML Formatting Tools are a group of SAS macros that
consist of (%OUT2HTM, %DS2HTM, and %TAB2HTM). The
collection of macros enable the programmer to format and
save SAS output for viewing in a WEB browser. The macros
assign valid HTML tags creating HTML files.  The HTML
files that are created are static pages which can be pushed
to the Intranet Server. Once on the Intranet server the SAS
output can be viewed. The macro %OUT2HTM is a great
formatting tool to get started in creating HTML pages.
PICTURES 1 and 2 are  HTML examples using
%OUT2HTM.

SOURCE CODE FOR PICTURES 1 and 2:

/* use FORMCHAR='|----|+|---+=|-/\<>*’ otherwise
characters that you use may look unusual depending on
your browser and the operating system used */

options ls=195 missing = '0' ps=100 nodate nonumber
 FORMCHAR='|----|+|---+=|-/\<>*';

data bccom;
set wext.bc wext.cdbc;
format nord 8.;
nord = prodord;
run;

data accom;
set wext.ac wext.cdac;
format nord 8.;
nord = prodord;
run;

proc sort data=bccom;
by acctbank grpord nord ;
run;

proc sort data=accom;
by acctbr grpord nord ;
run;

/* used for date in the title of the proc
report  */
data _null_;
x=&lastfri.;



call symput('dttitle',put(x,mmddyy10.));
run;

/* create proc format bank name commercial
data bktab1(keep = acctbank bankname);
set bccom(keep=acctbank bknam);
bkn = substr(bknam,1,3);

if bkn in ('STB') then do;
bankname = substr(bknam,5,40);
end;
else do;
bankname = bknam;
end;

run;

proc sort data=bktab1 nodupkey;
by acctbank bankname;run;

/* creates format for Bank names  */
data bankc;
  set bktab1(rename=(acctbank=start
bankname=label));
 fmtname='bankcom';
 type='C';
 keep fmtname label start type;
 run;

proc format cntlin=bankc;run;

***** branch section *****;
***** branch retail *****;

data branch(keep=acctbank acctbr brnam );
 set accom;
 run;

proc sort data=branch nodupkey;
by acctbank acctbr brnam;run;

/* very important step because it creates the
count needed for how many times the proc report
has to run – note the count and places names
for banks and branches */
data _null_;
 set branch end=no_more;
 call symput('ab'||left(_n_),acctbank);
 call symput('abn'||left(_n_),
                put(acctbank,    $bankcom.));
 call symput('br'||left(_n_),acctbr);
 call symput('brn'||left(_n_),brnam);
 if no_more then call symput('count',_n_);
run;

/* macro needed for the 3600 branches that
   Suntrust Banks have  */

%macro putloop;
   %local i;
   %do i=1 %to &count;

%macro
Branches(fdata,fname,brnam,hf,dlxls,namxls);

/* beginning of %OUT2HTM starts capturing
data */
%out2htm(capture=on);

/* note beginning proc report options
1. ls=195 the linesize max a report can
take for the web
2. nowindows is used so that the display
window for the proc report will not
produce an error during the running of
report */
proc report data=&fdata. split='*' ls=195
headline missing out=&dlxls. nowindows;
/* the basefont is used to make sure a
certain font is used to view the report */
title1 "<BASEFONT size=1.5>";

/* the following titles all use some HTML
in order to make sure the titles are
centered. Font size 3 is used for Letters
to appear bigger. The color blue is used
for the lettering */
title2 "<CENTER><FONT FACE=ARIAL SIZE=3
COLOR=BLUE>SUNTRUST - Bank -
&fname.</CENTER></FONT>";
title3 "<CENTER><FONT FACE=ARIAL SIZE=3
COLOR=BLUE>Branch - &&br&i
"&brnam."</CENTER></FONT>";
title4 "<CENTER><FONT FACE=ARIAL SIZE=3
COLOR=BLUE>Personal Deposits Open/Close
Weekly Report</CENTER></FONT>";
title5 "<CENTER><FONT FACE=ARIAL SIZE=3
COLOR=BLUE>Week Ending: &&dttitle.
</CENTER></FONT>";
title6 "<CENTER><FONT FACE=ARIAL SIZE=3
COLOR=BLUE>Note: All balances are in 000's
</CENTER></FONT>";

/* the following footnotes are hyperlinks
which are used in PICTURE2.
footnote1 is a hyperlink to a CSV file
which can be brought up in EXCEL.
footnote2 is a hyperlink to a USER GUIDE*/
footnote1 "<CENTER><FONT FACE=ARIAL SIZE=3
COLOR=BLUE><A HREF=./&namxls..CSV>DOWNLOAD
FILE</A></CENTER></FONT>";
footnote2 '<CENTER><FONT FACE=ARIAL SIZE=3
COLOR=BLUE><A
HREF="http://www.suntrust.com/WeeklyUserGu
ide.doc">HELP</A></CENTER></FONT>';

column grpord prodgrp prodroll onumcls
onumnew onumexs onumconv onumdcnv
       onewbal oexsbal oconvbal numcls
clsbal numnew newbal numexs
       exsbal numconv convbal numdconv
dconvbal
       ("_Total Accounts_" totaccts chg1b
totbals chg2b )
       ("_Opened Accounts_" numcon1
convbal1 numnew1 newbal1 chg3b)
       ("_Closed Accounts_" numdcon1
dconvbl1 numcls1 clsbal1 chg4b) ;
define grpord /order   noprint;
define prodgrp /order   noprint;
define prodroll / display format= $20.
width=20 'Deposits' left ;



define onumcls / sum noprint;
define onumnew / sum noprint;
define onumexs / sum noprint;
define onumconv / sum noprint;
define onumdcnv / sum noprint;
define onewbal / sum noprint;
define oexsbal / sum noprint;
define oconvbal / sum noprint;
define numcls  / sum noprint;
define clsbal / sum noprint;
define numnew  / sum noprint;
define newbal / sum noprint;
define numexs  / sum noprint;
define exsbal / sum noprint;
define numconv  / sum noprint;
define convbal / sum noprint;
define numdconv  / sum noprint;
define dconvbal / sum noprint;
define totaccts / computed format=comma10.
width=10 "**Number" right;
define chg1b / computed format=percent7.1 width=8
"% Change*over*prior*week";
define totbals / computed format=dollar14.
width=14 "**Balances" right;
define chg2b / computed format=percent7.1 width=8
"% Change*over*prior*week";
define numcon1  / computed format=comma11.
width=11 '# Converted' right;
define convbal1 / computed format=dollar10.
width=10 "Converted*Balances" right;
define numnew1  / computed format=comma9. width=9
"# New";
define newbal1 / computed format=dollar10.
width=10 "New *Balances";
define chg3b / computed format=percent9.1
width=12 "% Change in*Total New*over*prior week";
define numdcon1  / computed format=comma13.
width=13 '# Deconverted' right;
define dconvbl1 / computed format=dollar11.
width=11 "Deconverted*Balances" right;
define numcls1  / computed format=comma9. width=9
"# Closed";
define clsbal1 / computed format=dollar10.
width=10 "Closed*Balances";
define chg4b / computed format=percent9.1
width=12 '% Change in*Total Closed*over*prior
week';

compute totaccts;
totaccts = numnew.sum + numexs.sum + numconv.sum;
endcomp;

compute totbals;
totbals = (newbal.sum + exsbal.sum +
convbal.sum)/1000;
endcomp;

compute numcon1;
numcon1 = numconv.sum;
endcomp;

compute convbal1;
convbal1 = (convbal.sum)/1000;
endcomp;

compute numnew1;
numnew1 = numnew.sum;
endcomp;

compute newbal1;
newbal1 = (newbal.sum)/1000;
endcomp;

compute numdcon1;
numdcon1 = numdconv.sum;
endcomp;

compute dconvbl1;
dconvbl1 = (dconvbal.sum)/1000;
endcomp;

compute numcls1;
numcls1 = numcls.sum;
endcomp;

compute clsbal1;
clsbal1 = (clsbal.sum)/1000;
endcomp;

compute chg1b;
if (onumnew.sum + onumexs.sum +
onumconv.sum) > 0 then do;
chg1b = (numnew.sum + numexs.sum +
numconv.sum - onumnew.sum - onumexs.sum -
onumconv.sum)/(onumnew.sum + onumexs.sum +
onumconv.sum);
end;
else do;
chg1b = 0;
end;
endcomp;

compute chg2b;
if (onewbal.sum + oexsbal.sum +
oconvbal.sum) > 0 then do;
chg2b = (newbal.sum + exsbal.sum +
convbal.sum - onewbal.sum - oexsbal.sum -
oconvbal.sum)/(onewbal.sum + oexsbal.sum +
oconvbal.sum);
end;
else do;
chg2b = 0;
end;
endcomp;

compute chg3b;
if (onumnew.sum ) > 0 then do;
chg3b = (numnew.sum -
onumnew.sum)/(onumnew.sum);
end;
else do;
chg3b = 0;
end;
endcomp;

compute chg4b;
if (onumcls.sum) > 0 then do;
chg4b = (numcls.sum -
onumcls.sum)/(onumcls.sum);
end;
else do;
chg4b = 0;
end;
endcomp;

break before prodgrp / ;
break after prodgrp / ol skip summarize ;

compute before prodgrp;
line @1  prodgrp  $20. ;
endcomp;

compute after prodgrp;
prodroll = trim(prodgrp)||' Sub-Total';
endcomp;



rbreak after  / ol skip summarize ;

compute after;
prodroll = 'Total';
endcomp;
run;

/* the final 4 lines of %out2htm.
capture=off – stops capturing output.
encode=n lets the browser render the title with
attributes specified.
htmlfile – lets you name the HTML file.
ttag= no formatting – tells the output formatter
not to assign any HTML tags.
** you could also add bgtype=image which
indicates an image as a background type, then
specify the name of image in the BG argument.
*/
%out2htm(capture=off,
         encode=n,

     htmlfile=/gs/&hf. ,
         ttag=NO FORMATTING);

/* this section is used to rename the variables
that will be used for a CSV file which can be
displayed in EXCEL */

data b&dlxls.(drop= totaccts chg1b totbals chg2b
numcon1 convbal1 numnew1 newbal1 chg3b
                    numdcon1 dconvbl1 numcls1
clsbal1 chg4b);
set &dlxls.(drop =grpord onumcls onumnew onumexs
onumconv onumdcnv
       onewbal oexsbal oconvbal numcls clsbal
numnew newbal numexs
       exsbal numconv convbal numdconv dconvbal
_break_);
if prodroll in ('  ') then delete;
total_num_accts      = totaccts;
tot_acct_perc_chg    = chg1b;
total_balances       = totbals;
tot_bal_perc_chg     = chg2b;
num_converted        = numcon1;
converted_balances   = convbal1;
num_new              = numnew1;
new_balances         = newbal1;
new_perc_chg         = chg3b;
num_deconverted      = numdcon1;
deconverted_balances = dconvbl1;
num_closed           = numcls1;
closed_balances      = clsbal1;
closed_perc_chg      = chg4b;
run;

/* the creation of the CSV file used
   in the HYPERLINK on the HTML page. */
PROC EXPORT DATA= b&dlxls.
            OUTFILE="/gs/&namxls..csv"
            DBMS=CSV REPLACE;

RUN;

/* macro branches ending - which is used for
selection of acctbank and branch, name of
acctbank, name of branch, HTML file name, and
CSV file name */
%mend branches;
%branches(accom(where=(acctbank  in("&&ab&i")and
acctbr in ("&&br&i"

))),&&abn&i,"&&brn&i",RET&&fdt.Dep&&ab&i.&
&br&i...html,t1,RET&&fdt.Dep&&ab&i.&&br&i.
);
%end;

%mend putloop;
%putloop;

The above code creates the following HTML code:

which appears on the web like this:

The hyperlinks are at the bottom of each report:



Once the download file hyperlink is clicked it
appears as:

Select the ‘Open this file from its current
location’ option and it appears as:

Once the HELP hyperlink is clicked it
 appears as a word document:

CONCLUSION

SAS version 8 provides an excellent tool for producing Proc
Reports on the WEB. The %OUT2HTM macro is a very
helpful HTML formatting tool that is beneficial to Suntrust
Bank. Quite easy to use, 5 lines of codes can convert a
simple Report to an HTML page with similar characteristics
as the non-HTML type report. Remember, to first write the
PROC REPORT then add the SAS macro %OUT2HTM .
Next, add some HTML to your own specifications for your
desired results.

